!
! SYS_SIZE is the number of clicks (16 bytes) to be loaded.
! 0x3000 is 0x30000 bytes = 196kB, more than enough for current
! versions of linux
!
SYSSIZE = 0x3000
!
!	bootsect.s		(C) 1991 Linus Torvalds
!
! bootsect.s is loaded at 0x7c00 by the bios-startup routines, and moves
! iself out of the way to address 0x90000, and jumps there.
!
! It then loads 'setup' directly after itself (0x90200), and the system
! at 0x10000, using BIOS interrupts. 
!
! NOTE! currently system is at most 8*65536 bytes long. This should be no
! problem, even in the future. I want to keep it simple. This 512 kB
! kernel size should be enough, especially as this doesn't contain the
! buffer cache as in minix
!
! The loader has been made as simple as possible, and continuos
! read errors will result in a unbreakable loop. Reboot by hand. It
! loads pretty fast by getting whole sectors at a time whenever possible.

.globl begtext, begdata, begbss, endtext, enddata, endbss
.text
begtext:															!BIOS将bootsect.s读到0x7c00，然后跳到0x7c00处执行
.data																!bootsect.s 将自己从0x7c00读到0x90000(576KB)
begdata:															!			将setup.s读到0x90200(576.5KB)
.bss																!			将system读到0x10000(64KB)  然后跳到0x90000处执行
begbss:
.text																! kernel = boot(bootsect.s+setup.s) + system(head.s+main.c+...)
																	! 
SETUPLEN = 4				! nr of setup-sectors					! setup占4个扇区
BOOTSEG  = 0x07c0			! original address of boot-sector    	! BIOS上电后，首先将bootsect.s(占512字节/1个扇区)加载到内存地址0x07c0:0x0000处
INITSEG  = 0x9000			! we move boot here - out of the way	! bootsect.s首先将自己从0x7c00(31KB)移动到0x90000(576KB)处
SETUPSEG = 0x9020			! setup starts here						! 然后将sectup.s放在0x90200(576KB+512字节)处
SYSSEG   = 0x1000			! system loaded at 0x10000 (65536).		! system(head.s+main.c+...)被加载到0x1000:0(64KB)处
ENDSEG   = SYSSEG + SYSSIZE		! where to stop loading

! ROOT_DEV:	0x000 - same type of floppy as boot.
!		0x301 - first partition on first drive etc
ROOT_DEV = 0x306

entry _start
_start:
	mov	ax,#BOOTSEG		! 1) 这部分代码是将bootsect.s从0x07c00移动到0x90000
	mov	ds,ax
	mov	ax,#INITSEG
	mov	es,ax
	mov	cx,#256			!256次
	sub	si,si			! ds:si=0x07c00
	sub	di,di			! es:di=0x90000
	rep					! 重复执行256次
	movw				! 每次一个word(2字节) 256*2=512字节
	jmpi	go,INITSEG  !跳转到INITSEG:go=0x9000:go处继续执行(其实和没jmp前的执行顺序一致，都是继续执行从go:开始的语句)(jmpi是间接跳转)
go:	mov	ax,cs			! cs=0x9000
	mov	ds,ax			! ds=0x9000
	mov	es,ax			! es=0x9000
! put stack at 0x9ff00.
	mov	ss,ax			! ss=0x9000
	mov	sp,#0xFF00		! arbitrary value >>512 !sp=

! load the setup-sectors directly after the bootblock.
! Note that 'es' is already set up.

load_setup: 			! 2) 载入setup模块(占4个扇区)
	mov	dx,#0x0000		! drive 0, head 0
	mov	cx,#0x0002		! sector 2, track 0				! 从第2个扇区开始读(因为第一个扇区是boot loader)			
	mov	bx,#0x0200		! address = 512, in INITSEG		! 0x200=512字节 --> 读到0x90200
	mov	ax,#0x0200+SETUPLEN	! service 2, nr of sectors 	! ah=0x2表示读磁盘 al=SETUPLEN=x表示读入4个扇区
	int	0x13			! read it						! BIOS的0x13号中断: 读磁盘扇区
	jnc	ok_load_setup		! ok - continue
	mov	dx,#0x0000
	mov	ax,#0x0000		! reset the diskette			! 复位
	int	0x13
	j	load_setup		! 重读

ok_load_setup:			! 载入setup模块

! Get disk drive parameters, specifically nr of sectors/track

	mov	dl,#0x00
	mov	ax,#0x0800		! AH=8 is get drive parameters ! ah=8获取磁盘参数
	int	0x13
	mov	ch,#0x00
	seg cs
	mov	sectors,cx
	mov	ax,#INITSEG
	mov	es,ax

! Print some inane message	! 3) 打印开机logo

	mov	ah,#0x03		! read cursor pos
	xor	bh,bh
	int	0x10			! 读光标(获取光标位置)
	
	mov	cx,#48											!48个字符
	mov	bx,#0x0007		! page 0, attribute 7 (normal)	!7是显示属性
	mov	bp,#msg1		! 参数: 字符串的首地址
	mov	ax,#0x1301		! write string, move cursor
	int	0x10			! BISO的0x10号中断: 显示字符 	   ! 打印"Loading system..."

! ok, we've written the message, now		! 4)  加载system模块(os的其他代码)
! we want to load the system (at 0x10000) 	! 将system加载到0x10000(64KB)处

	mov	ax,#SYSSEG
	mov	es,ax		! segment of 0x010000
	call	read_it
	call	kill_motor

! After that we check which root-device to use. If the device is
! defined (!= 0), nothing is done and the given device is used.
! Otherwise, either /dev/PS0 (2,28) or /dev/at0 (2,8), depending
! on the number of sectors that the BIOS reports currently.

	seg cs
	mov	ax,root_dev
	cmp	ax,#0
	jne	root_defined
	seg cs
	mov	bx,sectors
	mov	ax,#0x0208		! /dev/ps0 - 1.2Mb
	cmp	bx,#15
	je	root_defined
	mov	ax,#0x021c		! /dev/PS0 - 1.44Mb
	cmp	bx,#18
	je	root_defined
undef_root:
	jmp undef_root
root_defined:
	seg cs
	mov	root_dev,ax

! after that (everyting loaded), we jump to
! the setup-routine loaded directly after
! the bootblock:
						! 跳到setup.s执行
	jmpi	0,SETUPSEG	! 最后，bootsect.s将控制权交给setup.s    cs:ip = SETUPSEG:0=0x90200

! This routine loads the system at address 0x10000, making sure
! no 64kB boundaries are crossed. We try to load it as fast as
! possible, loading whole tracks whenever we can.
!
! in:	es - starting address segment (normally 0x1000)
!
sread:	.word 1+SETUPLEN	! sectors read of current track
head:	.word 0			! current head
track:	.word 0			! current track

read_it:				!读入system(os的其他代码)
	mov ax,es
	test ax,#0x0fff
die:	jne die			! es must be at 64kB boundary
	xor bx,bx		! bx is starting address within segment
rp_read:
	mov ax,es
	cmp ax,#ENDSEG		! have we loaded all yet?
	jb ok1_read			! ENDSEG=SYSSEG+SYSSIZE SYSSIZE=0x8000  该变量可根据Image大小设定 ( 编译操作系统时 )
	ret
ok1_read:
	seg cs
	mov ax,sectors
	sub ax,sread		! sread 是当前磁道已读扇区数 ,ax未读扇区数
	mov cx,ax
	shl cx,#9
	add cx,bx
	jnc ok2_read
	je ok2_read
	xor ax,ax
	sub ax,bx
	shr ax,#9
ok2_read:
	call read_track		! 读磁道...
	mov cx,ax
	add ax,sread
	seg cs
	cmp ax,sectors
	jne ok3_read
	mov ax,#1
	sub ax,head
	jne ok4_read
	inc track
ok4_read:
	mov head,ax
	xor ax,ax
ok3_read:
	mov sread,ax
	shl cx,#9
	add bx,cx
	jnc rp_read
	mov ax,es
	add ax,#0x1000
	mov es,ax
	xor bx,bx
	jmp rp_read

read_track:
	push ax
	push bx
	push cx
	push dx
	mov dx,track
	mov cx,sread
	inc cx
	mov ch,dl
	mov dx,head
	mov dh,dl
	mov dl,#0
	and dx,#0x0100
	mov ah,#2
	int 0x13
	jc bad_rt
	pop dx
	pop cx
	pop bx
	pop ax
	ret
bad_rt:	mov ax,#0
	mov dx,#0
	int 0x13
	pop dx
	pop cx
	pop bx
	pop ax
	jmp read_track

!/*
! * This procedure turns off the floppy drive motor, so
! * that we enter the kernel in a known state, and
! * don't have to worry about it later.
! */
kill_motor:
	push dx
	mov dx,#0x3f2
	mov al,#0
	outb
	pop dx
	ret

sectors:
	.word 0

msg1:
	.byte 13,10											! 2字节
	.ascii "Loading liangwenhao0603@163.com system ..." ! 42字节("Loading system ...")
	.byte 13,10,13,10									! 4字节

.org 508	! 引导扇区的末尾 : BIOS用以识别引导扇区
root_dev:
	.word ROOT_DEV
boot_flag:
	.word 0xAA55 ! 扇区的最后两个字节必须是0xAA55     ! 是为了标识boot loader的结束位置

.text
endtext:
.data
enddata:
.bss
endbss:
